NPS55-77-37

NAVAL POSTGRADUATE SCHOOL

Monterey, California

AN INTERACTIVE MODEL TO COMPUTE

THE OFFICER MANPOWER PLAN

FOR THE UNITED STATES MARINE CORPS

by

Kneale T. Marshall

August 1977

Prepared for:
Headquarters Marine Corps
Code MPI20
Washington, DC 20380

NAVAL POSTGRADUATE SCHOOL
Monterey, California

Rear Admiral Isham Linder
Superintendent

Jack R. Borsting
Provost

This work was supported by the Manpower Planning Division (MPI20) of the Marine Corps through the Navy Personnel Research and Development Lab, San Diego.

Prepared by:

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE *(When Data Entered)*

REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

1. REPORT NUMBER: NPS55-77-37	2. GOVT ACCESSION NO.
3. RECIPIENT'S CATALOG NUMBER	
4. TITLE *(and Subtitle)*: An Interactive Model to Compute the Officer Manpower Plan for the United States Marine Corps	5. TYPE OF REPORT & PERIOD COVERED: Technical Report
6. PERFORMING ORG. REPORT NUMBER	
7. AUTHOR(s): Kneale T. Marshall	8. CONTRACT OR GRANT NUMBER(s)
9. PERFORMING ORGANIZATION NAME AND ADDRESS: Naval Postgraduate School, Monterey, CA 93940	10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS: N68221 77 WR 70052
11. CONTROLLING OFFICE NAME AND ADDRESS: Headquarters Marine Corps, Code MPI20, Washington, DC 20380	12. REPORT DATE: August 1977
	13. NUMBER OF PAGES: 42
14. MONITORING AGENCY NAME & ADDRESS *(if different from Controlling Office)*	15. SECURITY CLASS. *(of this report)*: UNCLASSIFIED
	15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT *(of this Report)*

Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT *(of the abstract entered in Block 20, if different from Report)*

18. SUPPLEMENTARY NOTES

19. KEY WORDS *(Continue on reverse side if necessary and identify by block number)*

Manpower
Forecasting
Personnel
Planning

20. ABSTRACT *(Continue on reverse side if necessary and identify by block number)*

This report describes a mathematical model and a set of associated interactive *APL* functions which are used to calculate an officer manpower plan for the United States Marine Corps. This plan is used during the budget preparation cycle and represents the future forecasted and planned force structure by rank. A complete plan can be calculated and printed in less than five minutes. This is a considerable reduction from the times involved in hand calculations. This increase in speed allows the planning officers to try

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE
S/N 0102-014-6601

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE *(When Data Entered)*

many alternate plans in a short period of time, and do sensitivity analysi
on certain data elements such as loss rates.

TABLE OF CONTENTS

1. Introduction

This report describes a mathematical model and a set of associated interactive *APL* functions which are used to calculate an officer manpower plan for the United States Marine Corps. This plan is used during the budget preparation cycle and represents the future forecasted and planned force structure by rank.

A complete plan can be calculated and printed in less than five minutes. This is a considerable reduction from the times involved in hand calculations. This increase in speed allows the planning officers to try many alternate plans in a short period of time, and do sensitivity analysis on certain data elements such as loss rates.

A brief summary of the method of calculation is given here. Details are presented in section 3. End strengths (by month) are given for ranks General through Captain, as well as losses and gains to these ranks. From these the monthly numbers of promotions out of ranks Colonel through 1-st Lieutenant are calculated. For ranks Warrant Officer-1 through 2-nd Lieutenant, the numbers of monthly promotions are given, and thus the monthly end-strengths for these ranks are determined. Finally, once promotions into and out of 1-st Lieutenant have been calculated, the monthly end-strengths are calculated for that rank.

Section 2 contains the mathematical formulation of the flow model which is based on ideas presented in chapter 1 of reference [2]. Section 3 gives a detailed description of the order in which the calculations are made. Section 4 describes a set of data input functions written in *APL* and intended for interactive use in entering data. Section 5 describes a set of calculation and format functions written in *APL* and intended for calculation and display of the officer manpower plan. Readers unfamiliar with *APL* should see reference [1]. Finally, section 6 shows a sample use of the functions and displays a plan.

All the functions in this report are in a workspace called *OFFICERPLAN* in the

APL+ system of Scientific Time Sharing Corporation. They were prepared at the Naval Postgraduate School through constant communication with the staff in MPP at Headquarters, Marine Corps.

2. The Personnel Flow Model

In this section is described the mathematical flow model and how it is used to determine the officer manpower plan. The model is based on simple conservation of flow equations (see chapter 1 of reference [2] for details of flow models). Let the discrete planning time (typically the end of each month) be numbered $t = 0,1,2, \ldots,$ with $t = 0$ being the starting point. The time interval between times t and $t+1$ we call period $t+1$. At each time t we count the number in each rank and call it the stock level at time t. People are promoted, join, or leave between the accounting time points, and these are called flows in period t. Thus we can write

(1) Stock level at (t+1) = Stock level at t + Gains in (t+1) - Losses in (t+1) + Promotions In in (t+1) - Promotions Out in (t+1)

To express (1) mathematically we use the following notation. Let

$s_j(t)$ = stock level at time t in rank j

$g_j(t)$ = gains in period t into rank j

$\ell_j(t)$ = losses in period t from rank j

$x_j(t)$ = promotions in period t into rank j

$y_j(t)$ = promotions in period t out of rank j

The index j ranges over $1,2, \ldots, n$, where n is the number of ranks which are numbered in decreasing seniority. Thus rank 1 is the highest rank and rank n is the lowest. Equation (1) now becomes

$$(2) \quad s_j(t+1) = s_j(t) + g_j(t+1) - \ell_j(t+1) + x_j(t+1) - y_j(t+1),$$

$$j = 1,2, \ldots, n, \; t \geq 0.$$

Let $s(t)$, $g(t)$, $\ell(t)$, $x(t)$ and $y(t)$ represent the n-dimensional vectors for $s_j(t)$, $g_j(t)$, $\ell_j(t)$, $x_j(t)$ and $y_j(t)$ respectively. Then we can write (2) as

$$(3)\quad s(t+1) = s(t) + g(t+1) - \ell(t+1) + x(t+1) - y(t+1),\ t \geq 0.$$

Before this general model can be applied to the USMC officer corps, further explanation is required of what we precisely mean by ranks, and for each rank, the definition of gains and losses.

In the computer models described later in this report n has the value 11 and the ranks are:

Number (j)	Rank	
1	General	(GEN)
2	Colonel	(COL)
3	Lieutenant Colonel	(LCOL)
4	Major	(MAJ)
5	Captain	(CAPT)
6	First Lieutenant	(1LT)
7	Second Lieutenant	(2LT)
8	Warrant Officer-4	(CWO-4)
9	Warrant Officer-3	(CWO-3)
10	Warrant Officer-2	(CWO-2)
11	Warrant Officer-1	(WO-1)

For each rank, the sources of gains are:

Rank	Sources of Gains
GEN	None
COL-CAPT	Returns from reimbursable billets
1LT-2LT	Accessions into the commissioned officer corps
CWO-4 - CWO-2	None
WO-1	Accessions into the warrant officer corps

For each rank the sources of losses are:

Rank	Sources of Losses
GEN	Natural attrition and retirement
COL-MAJ	Natural attrition, retirement and movement to reimbursable billets
CAPT	Natural attrition, those released from active duty, and movement to reimbursable billets
1LT	Natural attrition and those released from active duty
2LT - WO-1	Natural attrition

Promotions into a given rank are of course related to promotions out of lower ranks. Since time periods are typically of one-month length, promotions of more than 1 rank in a time period are assumed not to occur. An exception occurs in promotions from warrant officer status to limit duty officer (LDO) in the rank of 1LT.

For the ranks GEN through CAPT and for CWO-4 through CWO-2 , the promotions into a rank are equal to the promotions out of the next lower rank. Thus

$$x_j(t) = y_{j+1}(t), \quad j = 1,2, \ldots, 5, \quad j = 8,9,10, \quad t \geq 1. \tag{4}$$

Note that $y_1(t)$ is zero (no promotions out of the highest rank).

Let $z_j(t)$ be the number in rank j promoted to 1LT as LDO's, for $j = 8,9,10$, $t \geq 1$. Then

$$x_6(t) = y_7(t) + z_8(t) + z_9(t) + z_{10}(t), \quad t \geq 1. \tag{5}$$

No promotions are possible into 2LT or WO-1, thus $x_7(t)$ and $x_{11}(t)$ are both zero.

Equation (3) can now be simplified by eliminating $x(t+1)$. Let A be an 11 x 11 matrix of 0's and 1's where

(i) the diagonal immediately above the main diagonal has all elements equal to 1 except for the one on row 6 which is 0,

(ii) all other elements of A are 0.

Let B be an 11 x 11 matrix of 0's and 1's where

(i) elements in columns 8, 9, and 10 on row 6 are equal to 1,

(ii) all other elements are equal to 0.

Let $z(t)$ be an 11-vector with $z_8(t)$, $z_9(t)$ and $z_{10}(t)$ in positions 8, 9, and 10 respectively, and all other elements equal to zero. Then from (4) and (5)

$$x(t) = Ay(t) + Bz(t), \quad t \geq 1. \tag{6}$$

Substituting this into (3) gives

$$s(t+1) = s(t) + g(t+1) - \ell(t+1) + (A-I)y(t+1) + Bz(t+1), \quad t \geq 0. \tag{7}$$

Here I represents an identity matrix. Equation (7) can now be used to determine the officer manpower plan.

3. The Calculation Procedure

For ranks GEN through CAPT (1-5) the end strengths, gains and losses are all given. Denote with a $\bar{}$ the system comprising the first 5 equations in (7). Since the 5 x 5 matrix taken from the upper left of B is a zero matrix, we have

$$\bar{s}(t+1) = \bar{s}(t) + \bar{g}(t+1) - \bar{\ell}(t+1) + (\bar{A}-\bar{I})\bar{y}(t+1), \quad t \geq 0. \tag{8}$$

The only unknowns in this equation are the elements of $\bar{y}(t)$ for $t \geq 1$. Thus we solve (8) to obtain

$$\bar{y}(t+1) = (I-\bar{A})^{-1}\,[\bar{s}(t) - \bar{s}(t+1) + \bar{g}(t+1) - \bar{\ell}(t+1)], \quad t \geq 0. \tag{9}$$

The solution of (9) gives the promotions out of ranks COL through 1LT for each period in the planning horizon.

For ranks 2LT to WO-1, the promotions out are given, together with the gains and losses. Denote with $\hat{}$ the system comprising the last 5 equations in (7). Then given the starting stocks we determine $\hat{s}(1)$ by

$$\hat{s}(1) = \hat{s}(0) + \hat{g}(1) - \hat{\ell}(1) + (\hat{A}-I)\hat{y}(1) + \hat{B}\hat{z}(1), \tag{10}$$

then $\hat{s}(2)$, $\hat{s}(3)$ etc. recursively through the planning horizon. The only remaining rank for which calculations have not been made is 1LT. For ranks above this, end-strengths are given and promotions calculated. For ranks below 1LT, promotions are given and end-strengths calculated. For 1LT we have

$$s_6(t+1) = s_6(t) + g_6(t+1) - \ell_6(t+1) + x_6(t+1) - y_6(t+1), \quad t \geq 0.$$

Starting at $t = 0$, $s_6(0)$ is given, as are $g_6(1)$ and $\ell_6(1)$. Promotions out, $y_6(1)$, are calculated through equation (9), and promotions in, $x_6(1)$, are calculated through (10) and (5). Thus the unknown is the end-strength $s_6(1)$. This and future period end strengths are calculated recursively over the planning horizon after equations (9) and (10) have been solved for all other ranks.

4. *APL* Data Input Functions

The following page lists eleven matrices and one vector which are used to store input data required in computing the officer manpower plan. The results of the calculations are also stored in these arrays.

There are eight *APL* input functions used to interactively enter data at a remote terminal. Each one is described on the following pages together with a listing of the function and an illustrative example of its use. Alternative methods of data input are illustrated. The use of the *APL* operator ρ can save time when repetitive numbers have to be entered. See Gilman and Rose [1] for details of *APL* operators.

DATA ARRAYS

These arrays are stored as global variables.

APL Name	Dimensions	Description
BSM	11 x 12	Matrix of beginning strengths, 11 ranks (GEN - WO-1), 12 periods
ESM	11 x 12	Matrix of end strengths, 11 ranks, 12 periods
LFM	11 x 12	Matrix of loss factors, 11 ranks, 12 periods
LM	11 x 12	Matrix of losses, 11 ranks, 12 periods
MGM	11 x 12	Matrix of monthly gains, 11 ranks, 12 periods
POM	11 x 12	Matrix of promotions "out", 11 ranks 12 periods
LDOM	11 x 12	Matrix of promotions to LDO from warrant officer ranks, 11 ranks, 12 periods
ARM	11 x 12	Matrix of numbers in reimbursable billets, 11 ranks, 12 periods
TRM	11 x 12	Matrix of transfers to reimbursable billets, 11 ranks, 12 periods
FRM	11 x 12	Matrix of transfers from reimbursable billets, 11 ranks, 12 periods
RLM	11 x 12	Matrix of numbers released from active duty, 11 ranks, 12 periods
TLV	11	Vector of yearly losses from each rank, 11 ranks

Type)*SAVE* after data entry completion to ensure that new input data is not lost.

BSINPUT (begin strength input)

Syntax: Niladic, interactively asks for input

Global Variables: It requires the error message *EM*, the rank vector *RANKV*, and the data matrix *BSM*.

Files: None

Functions: None

Purpose: This function allows the user to enter the numbers in each of 11 ranks (GEN - WO-1) at the beginning of a planning period and store them as the first column of an 11 x 12 *BSM*.

Example:

```
      BSINPUT
BEGIN STRENGTH?
   GEN   COL  LCOL   MAJ  CAPT   1LT   2LT   W-4   W-3   W-2   W-1
□:
   67   573  1471  2909  4848  4086  3381   199    79   411   504
```

Function Listing:

```
      ∇BSINPUT[□]∇
    ∇ BSINPUT;X
[1]   L1:'BEGIN STRENGTH?' ◇ RANKV
[2]     →(11=ρX←□)ρL2 ◇ EM ◇ →L1
[3]   L2:BSM[;1]←X
    ∇
```

ESINPUT (end strength input)

Syntax: Niladic, interactively asks for input

Global Variables: It requires the error message *EM*, the month matrix *MONTH*, the rank vector *RANKV*, and the data matrix *ESM*.

Files: None

Functions: None

Purpose: This function allows the user to enter the planned end strengths for ranks GEN, COL, LCOL, MAJ, and CAPT for each of the 12 planning months. These are stored in the first 5 rows of the 11 x 12 matrix *ESM*.

Example:

```
      ESINPUT
END STRENGTH?
   GEN
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
⍞:
 67  67  67  67  67  67  67  67  67  67  67  67
   COL
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
⍞:
      12⍴574
  LCOL
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
⍞:
      12⍴1472
   MAJ
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
⍞:
      (5⍴2910),2912,6⍴2913
  CAPT
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
⍞:
      12⍴4848
```

Function Listing:

```
        ∇ESINPUT[⍞]∇
      ∇ ESINPUT;X
[1]     'END STRENGTH?' ◇ I←0
[2]   L1:RANKV[(6×I)+⍳6] ◇ ,MONTH,' '
[3]     →(12=⍴X←⍞)⍴L2 ◇ EM ◇ →L1
[4]   L2:ESM[I+1;]←X ◇ →(4≥I←I+1)⍴L1
      ∇
```

LFINPUT (loss factor input)

Syntax: Niladic, interactively asks for input

Global Variables: It requires the error message *EM*, the month matrix *MONTH*, the rank vector *RANKV* and the data matrix *LFM*.

Files: None

Functions: None

Purpose: This function allows the user to enter the desired loss factors for each of 11 ranks (starting with GEN and ending with WO-1) for each of the 12 planning months. The 12 month factors must add to 1.0. These factors are used to spread yearly losses from each rank by month. They are stored in the 11 x 12 matrix *LFM*.

Example:

```
      LFINPUT
LOSS FACTORS?
   GEN
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
 .04 .03 .02 .01 .06 .06 .06 .04 .05 .077

INCORRECT INPUT
   GEN
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
.04 .03 .02 .01 .06 .06 .05 .05 .07 .31 .1  .2
   COL
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
    '
    '
    '
    '

   W-1
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
.09 .16 .09 .05 .07 .04 .02 .01 .02 .07 .22 .16
```

Function Listing:

```
    ∇LFINPUT[□]∇
∇ LFINPUT;I;X
[1]   'LOSS FACTORS?' ◇ I←1
[2]  L1:RANKV[(6×I-1)+⍳6] ◇ ,MONTH.' '
[3]   →(12=⍴X←□)⍴L2 ◇ EM ◇ →L1
[4]  L2:LFM[I;]←X ◇ →(11≥I←I+1)⍴L1
∇
```

MGINPUT (monthly gains input)

Syntax: Niladic, interactively asks for input

Global Variables: It requires the error message *EM*, the rank vector *RANKV*, the month matrix *MONTH*, and the data matrix *MGM*.

Files: None

Functions: None

Purpose: This function allows the user to enter the expected gains in each of 12 months into the ranks WO-1, 2LT and 1LT respectively. These are stored in rows 11, 7 and 6 of the 11 x 12 matrix *MGM*.

Example:

```
      MGINPUT
 MONTHLY GAINS?
   W-1
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
  0   0   0   0 250   0   0   0   0   0   0   0
   2LT
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
242 152 158 102   0  19 134 173 363 140 288   0
   1LT
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
  2   1   3   2   6   0   0   2   0   3   0   5
```

Function Listing:

```
    ∇MGINPUT[□]∇
∇ MGINPUT;I;M;V;X
[1]   ' MONTHLY GAINS?' ◇ I←1 ◇ M← 3 6 ⍴(60+⍳6),(36+⍳6),30+⍳6 ◇ V← 11 7
[2]  L2:RANKV[,M[I;]] ◇ ,MONTH,' '
[3]   →(12=⍴X←□)⍴L3 ◇ EM ◇ →L2
[4]  L3:MGM[V[I];]←X ◇ →(3≥I←I+1)⍴L2
∇
```

POINPUT (promotions-out input)

Syntax: Niladic, interactively asks for input

Global Variables: It requires the error message *EM*, the rank vector *RANKV*, the month matrix *MONTH*, and the two data matrices *POM* and *LDOM*.

Files: None

Functions: None

Purpose: This function allows the user to enter the planned promotions in each of 12 months out of ranks 2LT, and the warrant officers. These are stored in rows 7 through 11 of the 11 x 12 matrix *POM*. The function then asks for the numbers in each 12 months who became limited duty officers (LDO) from each of the 4 warrant ranks. These numbers are stored in rows 8 through 11 of the 11 x 12 matrix *LDOM*.

Example:

```
      POINPUT
PROMOTIONS OUT?
   2LT
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
  0 415  14 194   0   4 115  60 750 331   9  14
   W-4
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
  12ρ0
   W-3
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
  7   4   0   0   0   0   0   0  39   0   0   0
   W-2
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
  0   0   0   0  66   0   0   0   0   0   0   0
   W-1
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
      (4ρ0),234,7ρ0
 TO LDO?
   W-4
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
  0   0   0   0   0   0   0   0   2   0   0   0
   W-3
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
      (8ρ0),25,(3ρ0)
```

```
      W-2
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
      (8ρ0),73,3ρ0
      W-1
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
      12ρ0
```

Function Listing:

```
      ∇POINPUT[□]∇
    ∇ POINPUT;I;M;X
[1]   'PROMOTIONS OUT?' ◇ I←1
[2]  L1:RANKV[(30+I×6)+⍳6] ◇ ,MONTH,' '
[3]   →(12=ρX←□)ρL2 ◇ EM ◇ →L1
[4]  L2:POM[I+6;]←X ◇ →(5≥I←I+1)ρL1 ◇ I←1 ◇ ' TO LDO?'
[5]  L3:RANKV[(36+I×6)+⍳6] ◇ ,MONTH,' '
[6]   →(12=ρX←□)ρL4 ◇ EM ◇ →L3
[7]  L4:LDOM[I+7;]←X ◇ →(4≥I←I+1)ρL3
    ∇
```

REIMINPUT (reimbursable input)

Syntax: Niladic, interactively asks for input

Global Variables: It requires the error message *EM*, the rank vector *RANKV*, the month matrix *MONTH*, and the data matrices *TRM* and *FRM*.

Files: None

Functions: None

Purpose: This function allows the user to enter scores to and from reimbursable billets for each of 12 months for the ranks COL, LCOL, MAJ, and CAPT and store them in rows 2 through 5 of the 11 x 12 matrices *TRM* and *FRM* respectively.

Example:

```
      REIMINPUT
   COL TO REIMB
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
  1   1 .10ρ0
   COL FROM REIMB
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
  1.11ρ0
  LCOL TO REIMB
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
       '
       '
       '
       '
       '
       '

VALUE ERROR
□      12ρO
          ^
□:     12ρ0
  CAPT TO REIMB
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
  2   0   0   0   1   0   0   0   0   0   0   00

  CAPT FROM REIMB
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
  1   0   1   0   0   0   0   2   0   0   0   0
```

Function Listing:

```
      ∇REIMINPUT[□]∇
    ∇ REIMINPUT;M;X;I
[1]   M← 4 6 ρ6↓30↑RANKV ◇ I←1
[2]  L1:M[I;],' TO REIMB' ◇ .MONTH.' '
[3]   →(12=ρX←□)ρL2 ◇ EM ◇ →L1
[4]  L2:TRM[I+1;]←X
[5]  L3:M[I;],' FROM REIMB' ◇ .MONTH.' '
[6]   →(12=ρX←□)ρL4 ◇ EM ◇ →L3
[7]  L4:FRM[I+1;]←X
[8]   →(4≥I←I+1)ρL1
    ∇
```

RLDINPUT (released from active duty input)

Syntax: Niladic, interactively asks for input

Global Variables: It requires the error message *EM*, the rank vector *RANKV*, the month matrix *MONTH*, and the data matrix *RLM*.

Files: None

Functions: None

Purpose: This function allows the user to enter the numbers in each of 12 months who will be released from active duty in the ranks CAPT and 1LT. The data is stored in rows 5 and 6 of the 11 x 12 matrix *RLM*.

Example:

```
      RLDINPUT
RELAD?
  CAPT
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
  4  16  14  6    2   9   4   7   9   8   14 12
   1LT
OCT NOV DEC JAN FEB MAR APR MAY JUN JUL AUG SEP
□:
 55  55  90  41  35  62  28  48  90  48  62  76
```

Function Listing:

```
    ∇RLDINPUT[□]∇
∇ RLDINPUT;I;M;V;X
[1]   'RELAD?' ◇ I←1 ◇ M← 2 6 ρ(24+⍳6),(30+⍳6) ◇ V← 5 6
[2]  L1:RANKV[.M[I;]] ◇ ,MONTH,' '
[3]   →(12=ρX←□)ρL2 ◇ EM ◇ →L1
[4]  L2:RLM[V[I];]←X ◇ →(2≥I←I+1)ρL1
∇
```

TLINPUT (total loss input)

Syntax: Niladic, interactively asks for input

Global Variables: It requires the error message *EM*, the rank vector *RANKV*, and the data vector *TLV*.

Files: None

Functions: None

Purpose: This function allows the user to enter the losses from each of the 11 ranks for the planning period, and store them in the 11-vector *TLV*.

Example:

```
      TLINPUT
 TOTAL YEAR LOSSES?
   GEN   COL  LCOL   MAJ  CAPT   1LT   2LT   W-4   W-3   W-2   W-1
□:
    11    95   155   160   593   109    35    60    25     8     4
```

Function Listing:

```
      ∇TLINPUT[□]∇
    ∇ TLINPUT
[1]   L1:' TOTAL YEAR LOSSES?' ◇ RANKV
[2]   →(11=ρTLV←□)ρ0 ◇ EM ◇ →L1
    ∇
```

5. *APL* Calculation and Output Format Functions

The calculation, interactive changing, and formatting of the officer manpower plan require ten *APL* functions. These are listed on the following pages. Only one is specified by the user, namely *OFPLAN*. This function calls the other nine at appropriate points and they remain invisible to the terminal user.

OFPLAN (officer manpower plan)

Syntax: Niladic, interactively asks for various changes in variables and for the amount of detail to be printed.

Global Variables: The data arrays: *BSM*, *ESM*, *LFM*, *MGM*, *POM*, *LDOM*, *TRM*, *FRM*, *RLM*, and *TLV*.

Files: None

Functions: Requires the utilities *AYN* (answer yes or no) and *DATE*, the formatting functions *FMTI*, *ESREP*, *OF1E*, *OMPREP* and *ATREIMB*, and the calculation functions *CHGFM*, *CHGIM*, *PLAN1*.

Purpose: This is the main interactive function for obtaining the manpower plan. It is the only one (other than functions to input data) that the user specifies when calculating a plan.

Function Listing: See next page.

```
      ∇OFPLAN[□]∇
    ∇ OFPLAN
[1]  L5:→(~AYN 'DISPLAY LOSS FACTORS?')ρL1 ◇ FMTF LFM
[2]  L1:→(~AYN 'CHANGE LOSS FACTORS?')ρL2 ◇ LFM←CHGFM LFM
[3]  L2:→(~AYN 'DISPLAY END STRENGTHS?')ρL3 ◇ FMTI 5 12 ↑ESM
[4]  L3:→(~AYN 'CHANGE END STRENGTHS?')ρL7 ◇ ESM←CHGIM ESM
[5]  L7:→(~AYN 'CHANGE REIMB?')ρL4 ◇ 'TO REIMB' ◇ TRM←CHGIM TRM ◇ 'FROM REIMB' ◇ FRM←CHGIM FRM
[6]  L4:PLAN1 ◇ LF
[7]   '       USMC OFFICER PLAN                          DATE:-',DATE ◇ 2ρLF
[8]   '                    MONTHLY END STRENGTHS' ◇ LF ◇ ESREP ◇ 5ρLF ◇ →(~AYN '1-E TABLE?')ρL6
[9]   '                    OFFICER 1-E' ◇ LF ◇ OF1E ◇ 5ρLF ◇ →(~AYN 'MP PLAN?')ρL6
[10]  '                    OFFICER MANPOWER PLAN ' ◇ OMPREP ◇ 2ρLF ◇ ATREIMB
[11] L6:→(AYN 'CONTINUE?')ρL5
    ∇
```

CHGFM

Syntax: Monadic, *R ← CHGFM M*

Global Variables: It requires the error message *EM*.

Files: None

Functions: None

Purpose: It allows the user to change any elements in a fractional matrix, and checks that the new entries are consistent with row sums equal to 1.0. It is used only inside other functions.

Function Listing:

```
      ∇CHGFM[□]∇
    ∇ R←CHGFM M;X;Y
[1]   'ENTER ZERO FOR RANK IF NO MORE CHANGES'
[2]  L1:'RANK NUMBER?' ◇ →(0=X←□)ρL6 ◇ →(11≥X)ρL2 ◇ EM ◇ →L1
[3]  L2:'CURRENT' ◇ ,'F6.2' □FMT M[X;]
[4]  L3:'NEW?' ◇ →(1=+/Y←□)ρL4 ◇ 'DO NOT ADD TO 1.0' ◇ →L3
[5]  L4:→(12=ρY)ρL5 ◇ 'NEED 12 NUMBERS' ◇ →L3
[6]  L5:M[X;]←Y ◇ →L1
[7]  L6:R←M
    ∇
```

CHGIM

Syntax: Monadic, *R ← CHGIM M*

Global Variables: It requires the error message, *EM*.

Files: None

Functions: None

Purpose: It allows the user to change any elements in an integer matrix. It is used only inside other functions.

Function Listing:

```
      ∇CHGIM[□]∇
    ∇ R←CHGIM M;X;Y
[1]     'ENTER ZERO FOR RANK IF NO MORE CHANGES'
[2]   L1:'RANK NUMBER?' ◇ →(0=X←□)ρL6 ◇ →(11≥X)ρL2 ◇ EM ◇ →L1
[3]   L2:'CURRENT' ◇ ,'I6' □FMT M[X;]
[4]     'NEW?' ◇ →(12=ρY←□)ρL4 ◇ 'NEED 12 NUMBERS' ◇ →L3
[5]   L4:M[X;]←Y ◇ →L1
[6]   L6:R←M
    ∇
```

Syntax: Niladic

Global Variables: It requires all the data arrays, *BSM*, *ESM*, *LFM*, *LM*, *MGM*, *POM*, *LDOM*, *TRM*, *FRM*, *RLM*, and *TLV*.

Files: None

Functions: Requires the integer rounding function *ROUND*

Purpose: This function performs all the calculations for the manpower plan discussed in section 3.

Function Listing: See next page for both *PLAN1* and *ROUND*.

```
      ∇PLAN1[□]∇
    ∇ PLAN1;IV;I;V;PEM;NEM;LM;EV
[1]   LM←ROUND LFM×⍉ 12 11 ⍴TLV
[2]   BSM[IV;1+⍳11]←ESM[IV←⍳5;⍳11]
[3]   POM[1+IV;]←B+.×ESM[IV;]+LM[IV;]+RLM[IV;]-MGM[IV;]+BSM[IV;]+FRM[IV;]-TRM[IV;]
[4]   BSM[6;1+⍳11]←11↑ESM[6;]←B1+.×(12↑BSM[6;1])+MGM[6;]+POM[7;]+(+⌿LDOM)-POM[6;]+RLM[6;]+LM[6;]
[5]   M← ¯5 12 ↑MGM+((12⍴1)⊖POM)-RLM+LM+POM+LDOM
[6]   E←BSM[6+IV;1] ◇ I←1
[7]  L1:ESM[6+IV;I]←E←E+M[;I]
[8]   →(12≥I←I+1)⍴L1 ◇ BSM[6+IV;1+⍳11]←ESM[6+IV;⍳11]
    ∇
```

```
      ∇ROUND[□]∇
    ∇ R←ROUND M;EV;TM;N
[1]   N←⍴EV←(+/M)-+/TM←⌊0.5+M ◇ I←1
[2]  L1:TM[I;J]←TM[I;J←1↑⍒TM[I;]]+EV[I] ◇ →(N≥I←I+1)⍴L1 ◇ R←TM
    ∇
```

FMTF

Syntax: Monadic, *R ← FMTF M*

Global Variables: It requires the month matrix *MONTH* and the rank vector *RANKV*.

Files: None

Functions: None

Purpose: It formats the fractional matrix *M* with column and row headers. It is used only within other functions.

Function Listing:

```
      ∇FMTF[□]∇
    ∇ FMTF M
[1]   '  RANK',,(12 3 ρ' '),MONTH
[2]   (11 6 ρRANKV),'F6.2' □FMT M
    ∇
```

FMTI

Syntax: Monadic, *R ← FMTI M*

Global Variables: It requires the month matrix *MONTH* and the rank vector *RANKV*.

Files: None

Functions: None

Purpose: It formats the matrix *M* into integers with column and row headers. It is used only within other functions.

Function Listing:

```
      ∇FMTI[□]∇
    ∇ FMTI M;N
[1]   N←(⍴M)[1]
[2]   '  RANK',,(12 3 ⍴' '),MONTH
[3]   ((N,6)⍴RANKV),'BI6' □FMT M
    ∇
```

Syntax: Niladic

Global Variables: It requires the month matrix *MONTH*, the rank vector *RANKV*, and the data arrays *BSM* and *ESM*.

Files: None

Functions: None

Purpose: It formats the "End Strength Report" as part of the manpower plan.

Function Listing:

```
      ∇ESREP[□]∇
    ∇ ESREP;MYA;X
[1]   MYA←(+/BSM+ESM)÷24
[2]   ' BEGIN',' GRADE',,(,(12 3 ρ' '),MONTH),'   MYA'
[3]   ('I6' □FMT 11 1 ↑BSM),(11 6 ρRANKV),'I6' □FMT(ESM),MYA
[4]   '       TOTAL',,'I6' □FMT+⌿ESM
    ∇
```

OF1E

Syntax: Niladic

Global Variables: It requires the month matrix *MONTH* and the data arrays *BSM*, *ESM*, *RLM*, and *LM*.

Files: None

Functions: None

Purpose: It formats the "Officer 1E Table" as part of the manpower plan.

Function Listing:

```
      ∇OF1E[□]∇
    ∇ OF1E;X;Y;M;V
[1]   V←+⌿M←⍉ 6 12 ⍴(+⌿BSM),X,Y,((Y←+⌿LM)+X←+⌿RLM),(+⌿MGM),+⌿ESM
[2]   '                  |       LOSSES      |'
[3]   'MONTH      BEGIN    RELAD    OTHER    TOTAL    GAINS      END'
[4]   (' ',MONTH,' '),(12 2 ⍴' '),'BI8' □FMT M
[5]   'TOTAL  ',,'I8' □FMT(M[1;1],(4↑1↓V),M[12;6])
    ∇
```

OMPREP

Syntax: Niladic

Global Variables: It requires the month matrix *MONTH*, rank vector *RANKV*, the formatting headers *H*1, *H*1*G*, *H*2, *H*3, *H*4, *H*5, *H*6, *H*7, *H*8, *H*9, *H*10, *H*11, *H*12, *H*13, and *H*14, the line feed *LF*, and all the data arrays.

Files: None

Functions: None

Purpose: It formats the "Officer Manpower Plan," the main output showing the monthly details of a yearly plan for each rank.

Function Listing: See next page.

```
      ∇OMPREP[□]∇
    ∇ OMPREP;M;V;PI;TL;R;L;OL;G
[1]   '  RANK                  TOTAL',,(12 3 ρ' '),MONTH
[2]   M←BSM[;1],BSM,(+/TRM),TRM,(+/LM),LM,(+/POM),POM,(+/PI),(PI←(12ρ1)⊖POM),(+/FRM),FRM,
      ESM[;12],ESM
[3]   (4 6 ρ(12ρ' '),6↑RANKV),(4 3 ρ' '),H1G,'BI6' □FMT M1←(1 0 1 0 1 0 1)/[1] 7 13 ρM ◇ LF
[4]   (7 6 ρ(18ρ' '),6↑6↓RANKV),(7 3 ρ' '),H1,'BI6' □FMT M2← 7 13 ρM[2;] ◇ LF
[5]   (7 6 ρ(18ρ' '),6↑12↓RANKV),(7 3 ρ' '),H1,'BI6' □FMT M3← 7 13 ρM[3;] ◇ LF
[6]   (7 6 ρ(18ρ' '),6↑18↓RANKV),(7 3 ρ' '),H1,'BI6' □FMT M4← 7 13 ρM[4;] ◇ LF
[7]   TL←(R←(+/R),R←RLM[5;])+L←13↑26↓V←M[5;]
[8]   (9 6 ρ(24ρ' '),6↑24↓RANKV),(9 3 ρ' '),H2,'BI6' □FMT M5← 9 13 ρ(26↑V),R,L,TL,
      ¯52↑V ◇ LF
[9]   M←(IN,OUT,(39ρIN),(OUT←13ρ0),(IN←13ρ1))/M
[10]  TL←(R←(+/R),R←RLM[6;])+OL←13↑13↓V←M[6;]
[11]  M6← 9 13 ρ(13↑V),R,OL,TL,((+/G),G←MGM[6;]),(¯26↑¯13↓V),((+/L),L←+⌿LDOM),¯13↑V
[12]  (9 6 ρ(24ρ' '),6↑30↓RANKV),(9 3 ρ' '),H3,'BI6' □FMT M6 ◇ LF
[13]  (5 6 ρ(12ρ' '),6↑36↓RANKV),(5 3 ρ' '),H4,'BI6' □FMT M7← 5 13 ρ(26↑V),(G←(+/G),G←MGM[
      7;]),(¯13↑¯26↓V),¯13↑V←M[7;] ◇ LF
[14]  TO1← 1 13 ρM1[1;]+M2[1;]+M3[1;]+M4[1;]+M5[1;]+M6[1;]+M7[1;]
[15]  TO1←TO1,[1] M2[2;]+M3[2;]+M4[2;]+M5[2;]
[16]  TO1←TO1,[1] M1[2;]+M2[3;]+M3[3;]+M4[3;]+M5[5;]+M6[4;]+M7[2;]
[17]  TO1←TO1,[1] M6[5;]+M7[3;] ◇ TO1←TO1,[1] M6[8;]
[18]  TO1←TO1,[1] M2[4;]+M3[4;]+M4[4;]+M5[6;]+M6[6;]+M7[4;]
[19]  TO1←TO1,[1] M1[3;]+M2[5;]+M3[5;]+M4[5;]+M5[7;]+M6[7;]
[20]  TO1←TO1,[1] M2[6;]+M3[6;]+M4[6;]+M5[8;]
[21]  TO1←TO1,[1] M1[4;]+M2[7;]+M3[7;]+M4[7;]+M5[9;]+M6[9;]+M7[5;]
[22]  H5,H6,'BI6' □FMT TO1 ◇ LF
[23]  (5 9 ρ45↑(18ρ' '),'CWO-4'),H7,'BI6' □FMT M8← 5 13 ρ(26↑V),(¯13↑¯13↓V),((+/L),L←LDOM[
      8;]),¯13↑V←M[8;] ◇ LF
[24]  H9,(12 12 ρH8),'BI6' □FMT M9← 12 13 ρ(0 ¯13 ↓V),((+/L),L←LDOM[8+⍳2;]), 2 ¯13 ↑V←M[
      8+⍳2;] ◇ LF
[25]  M10← 6 13 ρ(26↑V),((+/G),G←MGM[11;]),(¯13↑¯26↓V),((+/L),L←LDOM[11;]),¯13↑V←M[
      11;]
[26]  (6 9 ρ54↑(18ρ' '),'WO-1'),H10,'BI6' □FMT M10 ◇ LF
[27]  TW←(M10[1 2 ;]+M8[1 2 ;]+M9[7 8 ;]+M9[1 2 ;]),[1] M10[3;] ◇ TW←TW,[1] M8[4;]+M10[
      5;]++⌿M9[5 11 ;]
[28]  TW←TW,[1] M10[4;]++⌿M9[3 9 ;] ◇ TW←TW,[1] M8[3;]++⌿M9[4 10 ;] ◇ TW←TW,[1] M8[
      5;]+M10[6;]++⌿M9[6 12 ;]
[29]  H12,H11,'BI6' □FMT TW ◇ LF ◇ H14,H13,'BI6' □FMT TO←TO1[1 2 3 4 8 9 ;]+(1 0 1 1 0 1)\[
      1] TW[1 2 3 7 ;]
    ∇
```

Syntax: Niladic

Global Variables: It requires the data arrays *ARM*, *TRM* and *FRM*.

Files: None

Functions: None

Purpose: It formats the movement of officers at, to, and from reimbursable billets as part of the manpower plan.

Function Listing:

```
∇ATREIMB[□]∇
     ∇ ATREIMB
 [1]    '          OFFICERS IN REIMBURSABLE BILLETS'
 [2]    '             COL       LCOL        MAJ       CAPT'
 [3]    'BEGIN',,'I10' □FMT,ARM[1+⍳4;1]
 [4]    '   TO',,'I10' □FMT(+/TRM)[1+⍳4]
 [5]    ' FROM',,'I10' □FMT(+/FRM)[1+⍳4]
 [6]    '  END',,'I10' □FMT(+/TRM+ARM-FRM)[1+⍳4]
     ∇
```

6. An Example

The next seven pages give an example of an officer manpower plan. This example does not show every feature of the interactive capability, but should be sufficient to illustrate the use of the model. A complete plan takes less than five minutes of terminal time and less than 1 second CPU time on Scientific Time Sharing's *APL*+ system. Thus many iterations are possible in the planning cycle.

```
      OFPLAN
DISPLAY LOSS FACTORS? YES
  RANK    OCT    NOV    DEC    JAN    FEB    MAR    APR    MAY    JUN    JUL    AUG    SEP
   GEN   0.04   0.03   0.02   0.01   0.06   0.06   0.05   0.05   0.07   0.31   0.10   0.20
   COL   0.07   0.05   0.03   0.05   0.07   0.05   0.04   0.05   0.08   0.25   0.14   0.12
  LCOL   0.10   0.06   0.04   0.05   0.08   0.06   0.05   0.05   0.06   0.18   0.14   0.13
   MAJ   0.08   0.06   0.05   0.07   0.07   0.05   0.06   0.05   0.10   0.16   0.14   0.11
  CAPT   0.07   0.07   0.13   0.07   0.06   0.11   0.07   0.07   0.08   0.09   0.09   0.09
   1LT   0.08   0.08   0.13   0.06   0.05   0.09   0.04   0.07   0.13   0.07   0.09   0.11
   2LT   0.08   0.08   0.08   0.07   0.05   0.09   0.08   0.10   0.10   0.08   0.11   0.08
   W-4   0.05   0.05   0.07   0.08   0.08   0.09   0.07   0.10   0.07   0.08   0.10   0.16
   W-3   0.11   0.06   0.07   0.07   0.10   0.08   0.07   0.06   0.08   0.10   0.08   0.12
   W-2   0.09   0.05   0.16   0.06   0.05   0.13   0.08   0.02   0.04   0.12   0.11   0.09
   W-1   0.09   0.16   0.09   0.05   0.07   0.04   0.02   0.01   0.02   0.07   0.22   0.16
CHANGE LOSS FACTORS? NO
DISPLAY END STRENGTHS? YES
  RANK    OCT    NOV    DEC    JAN    FEB    MAR    APR    MAY    JUN    JUL    AUG    SEP
   GEN     67     67     67     67     67     67     67     67     67     67     67     67
   COL    574    575    575    574    574    574    574    574    574    574    574    574
  LCOL   1472   1472   1472   1472   1472   1472   1472   1472   1472   1472   1472   1472
   MAJ   2910   2910   2910   2910   2910   2910   2910   2910   2910   2910   2910   2910
  CAPT   4848   4848   4848   4848   4848   4848   4848   4848   4848   4848   4848   4848
CHANGE END STRENGTHS? YES
ENTER ZERO FOR RANK IF NO MORE CHANGES
RANK NUMBER?
□:
      2
CURRENT
   574   575   575   574   574   574   574   574   574   574   574   574
NEW?
□:
      574 743 575
          574 575,9ρ574
RANK NUMBER?
□:
      0
CHANGE REIMB? NO
```

USMC OFFICER PLAN DATE:-20 AUGUST 1977

MONTHLY END STRENGTHS

BEGIN	GRADE	OCT	NOV	DEC	JAN	FEB	MAR	APR	MAY	JUN	JUL	AUG	SEP	MYA
67	GEN	67	67	67	67	67	67	67	67	67	67	67	67	67
573	COL	574	574	575	574	574	574	574	574	574	574	574	574	574
1471	LCOL	1472	1472	1472	1472	1472	1472	1472	1472	1472	1472	1472	1472	1472
2909	MAJ	2910	2910	2910	2910	2910	2910	2910	2910	2910	2910	2910	2910	2910
4848	CAPT	4848	4848	4848	4848	4848	4848	4848	4848	4848	4848	4848	4848	4848
4086	1LT	3942	4212	4018	4095	3992	3827	3841	3776	4432	4571	4383	4198	4103
3381	2LT	3620	3354	3495	3401	3399	3411	3427	3543	3152	2958	3233	3216	3358
199	W-4	203	204	200	195	190	185	181	175	208	203	197	187	195
79	W-3	72	66	64	62	125	123	121	119	53	50	48	45	80
401	W-2	399	399	398	398	566	565	564	564	491	490	489	488	481
504	W-1	504	502	502	502	518	518	518	518	518	518	517	516	512
	TOTAL	18611	18608	18549	18524	18661	18500	18523	18566	18725	18661	18738	18521	

1-E TABLE? YES

OFFICER 1-E

		LOSSES				
MONTH	BEGIN	RELAD	OTHER	TOTAL	GAINS	END
OCT	18518	59	95	154	244	18611
NOV	18611	71	85	156	153	18608
DEC	18608	104	116	220	161	18549
JAN	18549	47	82	129	104	18524
FEB	18524	37	82	119	256	18661
MAR	18661	71	109	180	19	18500
APR	18500	32	79	111	134	18523
MAY	18523	55	82	137	180	18566
JUN	18566	99	105	204	363	18725
JUL	18725	56	151	207	143	18661
AUG	18661	76	135	211	288	18738
SEP	18738	88	134	222	5	18521
TOTAL	18518	795	1255	2050	2050	18521

MP PLAN? YES

OFFICER MANPOWER PLAN

RANK		TOTAL	OCT	NOV	DEC	JAN	FEB	MAR	APR	MAY	JUN	JUL	AUG	SEP
	BEGIN	67	67	67	67	67	67	67	67	67	67	67	67	67
	LOSSES	11					1	1	1	1	1	3	1	2
GEN	PROM IN	11					1	1	1	1	1	3	1	2
	END	67	67	67	67	67	67	67	67	67	67	67	67	67
	BEGIN	573	573	574	574	575	574	574	574	574	574	574	574	574
	TO REIMB													
	LOSSES	95	7	5	3	5	7	5	4	5	8	22	13	11
COL	PROM OUT	11					1	1	1	1	1	3	1	2
	PROM IN	106	7	5	4	4	8	6	5	6	9	25	14	13
	FROM REIMB	1	1											
	END	574	574	574	575	574	574	574	574	574	574	574	574	574
	BEGIN	1471	1471	1472	1472	1472	1472	1472	1472	1472	1472	1472	1472	1472
	TO REIMB													
	LOSSES	155	16	9	6	8	12	9	8	8	9	28	22	20
LCOL	PROM OUT	106	7	5	4	4	8	6	5	6	9	25	14	13
	PROM IN	261	23	14	10	12	20	15	13	14	18	53	36	33
	FROM REIMB	1	1											
	END	1472	1472	1472	1472	1472	1472	1472	1472	1472	1472	1472	1472	1472
	BEGIN	2909	2909	2910	2910	2910	2910	2910	2910	2910	2910	2910	2910	2910
	TO REIMB													
	LOSSES	160	13	10	8	11	11	8	10	8	16	25	22	18
MAJ	PROM OUT	261	23	14	10	12	20	15	13	14	18	53	36	33
	PROM IN	421	36	24	18	23	31	23	23	22	34	78	58	51
	FROM REIMB	1	1											
	END	2910	2910	2910	2910	2910	2910	2910	2910	2910	2910	2910	2910	2910

	BEGIN	4848	4848	4848	4848	4848	4848	4848	4848	4848	4848	4848	4848	4848
	TO REIMB													
	RELAD	105	4	16	14	6	2	9	4	7	9	8	14	12
	OTHER LOSSES	593	42	42	76	42	36	65	42	42	47	53	53	53
CAPT	*TOTAL LOSSES*	698	46	58	90	48	38	74	46	49	56	61	67	65
	PROM OUT	421	36	24	18	23	31	23	23	22	34	78	58	51
	PROM IN	1119	82	82	108	71	69	97	69	71	90	139	125	116
	FROM REIMB													
	END	4848	4848	4848	4848	4848	4848	4848	4848	4848	4848	4848	4848	4848
	BEGIN	4086	4086	3942	4212	4018	4095	3992	3827	3841	3776	4432	4571	4383
	RELAD	690	55	55	90	41	35	62	28	48	90	48	62	76
	OTHER LOSSES	109	9	9	13	7	5	10	4	8	14	8	10	12
	TOTAL LOSSES	799	64	64	103	48	40	72	32	56	104	56	72	88
1LT	*GAINS*	24	2	1	3	2	6			2		3		5
	PROM OUT	1119	82	82	108	71	69	97	69	71	90	139	125	116
	PROM IN	1906		415	14	194		4	115	60	750	331	9	14
	FROM W.O.	100									100			
	END	4198	3942	4212	4018	4095	3992	3827	3841	3776	4432	4571	4383	4198
	BEGIN	3381	3381	3620	3354	3495	3401	3399	3411	3427	3543	3152	2958	3233
	LOSSES	35	3	3	3	2	2	3	3	2	4	3	4	3
2LT	*GAINS*	1776	242	152	158	102		19	134	178	363	140	288	
	PROM OUT	1906		415	14	194		4	115	60	750	331	9	14
	END	3216	3620	3354	3495	3401	3399	3411	3427	3543	3152	2958	3233	3216
	BEGIN	17335	17335	17433	17437	17385	17367	17262	17109	17139	17190	17455	17400	17487
	TO REIMB													
	LOSSES	1953	149	149	213	122	111	172	104	129	198	198	201	207
TOTAL	*GAINS*	1800	244	153	161	104	6	19	134	180	363	143	288	5
	FROM W.O.	100									100			
2LT - GEN	*PROM OUT*	3824	148	540	154	304	129	146	226	174	902	629	243	229
	PROM IN	3824	148	540	154	304	129	146	226	174	902	629	243	229
	FROM REIMB	3	3											
	END	17285	17433	17437	17385	17367	17262	17109	17139	17190	17455	17400	17487	17285

	BEGIN	199	199	203	204	200	195	190	185	181	175	208	203	197
	LOSSES	60	3	3	4	5	5	5	4	6	4	5	6	10
CWO-4	*PROM IN*	50	7	4							39			
	TO LDO	2									2			
	END	187	203	204	200	195	190	185	181	175	208	203	197	187
	BEGIN	79	79	72	66	64	62	125	123	121	119	53	50	48
	LOSSES	25		2	2	2	3	2	2	2	2	3	2	3
CWO-3	*PROM OUT*	50	7	4							39			
	PROM IN	66					66							
	TO LDO	25									25			
	END	45	72	66	64	62	125	123	121	119	53	50	48	45
	BEGIN	401	401	399	399	398	398	566	565	564	564	491	490	489
	LOSSES	8	2		1			1	1			1	1	1
CWO-2	*PROM OUT*	66					66							
	PROM IN	234					234							
	TO LDO	73									73			
	END	488	399	399	398	398	566	565	564	564	491	490	489	488
	BEGIN	504	504	504	502	502	502	518	518	518	518	518	518	517
	LOSSES	4		2									1	1
WO-1	*GAINS*	250					250							
	PROM OUT	234					234							
	TO LDO													
	END	516	504	502	502	502	518	518	518	518	518	518	517	516
	BEGIN	1183	1183	1178	1171	1164	1157	1399	1391	1384	1376	1270	1261	1251
	LOSSES	97	5	7	7	7	8	8	7	8	6	9	10	15
TOTAL	*GAINS*	250					250							
WARRANT	*TO LDO*	100									100			
OFFICERS	*PROM OUT*	350	7	4			300				39			
	PROM IN	350	7	4			300				39			
	END	1236	1178	1171	1164	1157	1399	1391	1384	1376	1270	1261	1251	1236

	BEGIN	18518	18518	18611	18608	18549	18524	18661	18500	18523	18566	18725	18661	18738
	TO REIMB													222
TOTAL	LOSSES	2050	154	156	220	129	119	180	111	137	204	207	211	5
OFFICERS	GAINS	2050	244	153	161	104	256	19	134	180	363	143	288	
	FROM REIMB		3	3										
	END	18521	18611	18608	18549	18524	18661	18500	18523	18566	18725	18661	18738	18521

OFFICERS IN REIMBURSABLE BILLETS

	COL	LCOL	MAJ	CAPT
BEGIN	8	21	9	2
TO	0	0	0	0
FROM	1	1	1	0
END	7	20	8	2

CONTINUE? NO_

REFERENCES

[1] Gilman, L. and Rose, A. J., APL: An Interactive Approach, J. Wiley, New York, 1974.

[2] Grinold, R. C. and Marshall, K. T., Manpower Planning Models, Elsevier-North Holland, 1977.